DISKRÉTNÍ MODELY

ÚVODNÍ PŘÍKLADY

ÚLOHA VÝROBNÍHO PLÁNOVÁNÍ

Cílem je obvykle určit objem výroby jednotlivých produktů tak, aby zisk byl maximální, případně náklady minimální, a to při respektování vstupních podmínek (suroviny, kapacita, čas) a výstupních podmínek (požadavky odběratelů, poměr, v němž se mají výrobky vyrábět…). Výrobky jsou obvykle celočíselnými proměnnými.

$z = cTx \rightarrow max$
$Ax \leq b$
$x \geq 0, x - celé$

```
!Babička plete čepice, šály a rukavice a pak je prodává. K dispozici má 50
klubek bílé vlny a 30 klubek modré vlny.
Na jednu čepici spotřebuje 3 klubka modré vlny a 1 klubko bílé vlny.
Na jednu šálu spotřebuje 2 klubka modré vlny a 2 klubka bílé vlny.
Na jedny rukavice spotřebuje jedno klubko modré vlny a jedno klubko bílé
vlny.
Čepici pak babička prodává za 250 Kč, šálu za 300 Kč a rukavice za 150 Kč.
Kolik má čeho uplést, aby její zisk byl maximální?;

!Proměnné: x1 = čepice, x2 = šály, x3 = rukavice.;

max=250*x1+300*x2+150*x3;
3*x1+2*x2+1*x3<=50;
1*x1+2*x2 + 1*x3<=30;
@gin(x1);  !omezení: jen celá čísla;
@gin(x2);
@gin(x3);
```

ŘEZNÁ ÚLOHA

Cílem je většinou rozdělit větší celky na menší, například nařezání desek na kratší prkna tak, aby počet rozřezaných desek byl co nejmenší, nebo aby odpad byl minimální. Omezujícími podmínkami je obvykle požadavek na počet nařezaných částí nebo jejich poměr. Je potřeba nejprve nalézt všechna řezná schémata: s nimi se pak v úloze pracuje.

```
!Máme tyče o délce jeden metr. Z nich je potřeba nařezat 20 tyčí o délce 50 cm,25 tyčí o délce 40 cm a 40 tyčí o délce 30 cm.
Určete, jak jakým způsobem bychom měli tyče rozřezat, aby
a) počet použitých tyčí byl co nejmenší
b) odpad po řezání byl minimální;

Existuje 6 řezných schémat:
```

schéma	**1**	**2**	**3**	**4**	**5**	**6**
50 cm	2	1	1	0	0	0
40 cm	0	1	0	2	1	0
30 cm	0	0	1	0	2	3
odpad	0	10	20	20	0	10

```
model:
sets:
schemata /1..6/:odpad,x; !odpad = kolik toho z tyče zbude při daném řezném schématu, x = kolikrát budeme řezat podle daného schématu;
vyrobky /1..3/:pozadavky;
tabulka(vyrobky,schemata):a; !a = kolik tyci dane delky ziskame danym schematem;
endsets

@for(vyrobky(i):@sum(schemata(j):a(i,j)*x(j))>=pozadavky(i));
min=ufce1; !min ufce 1 = snaha o co nejmensi pocet rozrezanych tyci. min ufce 2 = snaha o minimalizaci odpadu;
ufce1=@sum(schemata(i):x(i)); !minimalizujeme počet rozřezaných tyčí;
ufce2=@sum(schemata:odpad*x); !minimalizujeme odpad;
@for(schemata:@gin(x));

!Pokud bychom chtěli minimalizovat počet rozřezaných tyčí za podmínky, že odpad musí být nulový,použijeme dvoustupňovou metodu: zafixujeme ufce2 na nule a minimalizujeme ufce1.;
!ufce2=0;


data:
!a jsou řezná schémata, čili způsoby, jakými můžeme metrovou tyč nařezat na kratší tyče požadované délky.
Takových způsobů je 6. Například můžeme nařezat 2 tyče o délce 50 cm, odpad pak bude nulový. Nebo můžeme nařezat 1 tyč o délce 50 cm a jednu tyč o délce 40 cm, odpad pak bude 10 cm.;
a=
2 1 1 0 0 0
0 1 0 2 1 0
0 0 1 0 2 3;
pozadavky = 20 25 40; !kolik tyčí jednotlivých délek požadujeme;
odpad = 0 10 20 20 0 10;
enddata
end
```

ÚLOHA BATOHU (KNAPSACK PROBLEM)

Cílem je nalézt množinu projektů, která přinese maximální výnos, pokud jejich realizace bude financována z rozpočtu o velikosti ***K***. Zavádí se binární proměnné x_i, které se budou rovnat 1, pokud bude daný projekt vybrán, v opačném případě se budou rovnat 0. Výnos z projektu *i* označíme c_i, náklad na projekt i označíme a_i.

max:

$$z = \sum_{i=1}^{n} c_i x_i = c^T x$$

Snažíme se maximalizovat celkový výnos ze všech vybraných projektů.

$$a^T x \le K,$$

$$x \in B^n, B = \{0,1\}.$$

Náklad na vybrané projekty nesmí překročit rozpočet.
xi jsou binární proměnné, které nabývají 1 v případě, že je i-tý projekt vybrán

```
!Máme 5 projektů, z nichž každý je charakterizovaný náklady na realizaci a
výnosem. Výše investičních prostředků je 50 000 Kč, z kterých máme projekty
financovat a realizovat tak, aby vybrané projekty přinesly co nejvyšší
celkový výnos.
```

	P1	P2	P3	P4	P5
náklady ai	12000	10000	15000	18000	16000
výnosy ci	2000	18000	22000	26000	21000;

```
model:
sets:
projekt/P1,P2,P3,P4,P5/: x,naklady,vynosy;
endsets

data:
naklady = 12000 10000 15000 18000 16000;
vynosy = 20000 18000 22000 26000 21000;
rozpocet = 50000;
enddata

max = @sum(projekt: x*vynosy);
@sum(projekt: x*naklady) <=rozpocet;
@for(projekt: @bin(x));
end
Zdroj: Ing. Jan Fábry, Ph.D.: 4EK314 Diskrétní modely
```

ZDROJE:

Ing. J. Fábry, Ph.D.: přednášky 4EK314 Diskrétní modely, 2011.